2.5 Anwendungen der Newtonschen Gesetze

2.5.1 Bahn eines Tennisballs, der sich dreht

Die Reibungskraft der Luft kann leicht von der Größenordnung des Gewichtes des Körpers sein, der sich in ihr bewegt. Die Geschwindigkeit ist manchmal so groß, dass man die Reibungskraft der Luft als proportional zum Quadrat der Geschwindigkeit annehmen muss, vgl. (2.4-2)

$$F_r = \frac{1}{2} C A \rho v^2 \tag{2.5-1}$$

$\rho =$ Dichte der Luft. Für einen Baseball gilt annähernd $C = 0.3$.
Die beiden skalaren Bewegungsgleichungen sind

$$m\ddot{x} = -F_r \cos(\alpha) = -F_r \tfrac{v_y}{v}$$

$$m\ddot{y} = -mg - F_r \sin(\alpha) = -mg - F_r \frac{v_y}{v} \tag{2.5-2}$$

Die Anfangsgeschwindigkeit eines Baseballs kann 50 m/s erreichen. Erfahrene Baseballspieler verstehen es, einem Ball einen Drall (Effet) zu geben, wodurch der Ball einen zusätzlichen Auftrieb erhält (Magnus-Effekt).

Das Gleichungssystem (2.5-2) hat keine analytische Lösung, denn es handelt sich um zwei Differentialgleichungen zweiter Ordnung, die über v miteinander gekoppelt sind. Um sie numerisch zu lösen, wollen wir sie ein wenig abänden.

$$\ddot{x} = -rvv_x = -r\sqrt{v_x^2 + v_y^2} \cdot v_x$$

$$\ddot{y} = -g - rvv_y = -g - r\sqrt{v_x^2 + v_y^2} \cdot v_y \tag{2.5-3}$$

Die Widerstandszahl ist wieder definiert durch $r := \frac{1}{2m} C\rho A$.

Um jetzt auch die Drall-Kraft zu berücksichtigen, benutzen wir einen Ansatz, der (2.5-1) gleicht:

$$F_L = \frac{1}{2} C_L A \rho v^2 \tag{2.5-4}$$

Der Subindex L soll an Lift erinnern. Beide Koeffizienten C, C_L hängen von der Geschwindigkeit v, der Drehung und von der Oberflächenart ab. Im Allgemeinen muss man sie experimentell bestimmen. Die Kraft $\vec{F}_L$ steht senkrecht auf der Geschwindigkeit und der Rotationsachse des Balles.



Fig.2.5-1

Die beiden Koordinaten F_{Lx}, F_{Ly} sind wie folgt gegeben:

$$F_{Lx}/F_L = \cos(90-\alpha) = \sin(\alpha) = v_y/v \text{ und}$$
$$F_{Ly}/F_L = \sin(90-\alpha) = \cos(\alpha) = v_x/v$$

Jetzt haben wir alles zusammen, um ein MuPAD-Programm zu schreiben, in dem wir die Funktion `numeric::odessolve2` benutzen werden.

Im folgenden Bild zeigen wir nochmals die Kräfte.



Fig.2.5-2

Der Vektor $\vec{F}_L$ steht senkrecht auf $\vec{v}$, das in der x-y-Ebene bleibt. Mit ω (im Bild w) bezeichnen wir den Vektor der Drehung, den wir in der Rechnung in z-Richtung zeigen lassen. Hier ist zunächst das Programm, mit dem der fliegende Ball animiert gesehen wird. Im Anschluss folgen einige Erklärungen.

```
reset():
g:=9.81: rho:=1.29://Dichte der Luft
R:=0.037: m:=0.15://Radius und Kugelmasse
r1:=rho*R^2*PI/(2*m):
a:=60*PI/180://a = Abschusswinkel (60 Grad)
x0:=0:y0:=0://Anfangsposition
v0:=45://Anfangsgeschwindigkeit m/s
vx0:=v0*cos(a):// x-Koordinate von v0
vy0:=v0*sin(a):
//System von Diff.Gl. mit Anfangswerten
c:=0.3:cl:=0.0:
IVP1:={x''(t)=-r1*sqrt(x'(t)^2+y'(t)^2)*(c*x'(t)+cl*y'(t)),
y''(t)=-g+r1*sqrt(x'(t)^2+y'(t)^2)*(cl*x'(t)-c*y'(t)),
x(0)=x0,x'(0)=vx0,
y(0)=y0,y'(0)=vy0}:\newline{}
//Liste der Loesungen("fields"genannt)
fields:=[x(t),y(t),x'(t),y'(t)]:
ivp1:=numeric::ode2vectorfield(IVP1, fields):
Y1 := numeric::odesolve2(ivp1): Y1(6.535);//x,y,x',y' nach 6.535s
//[96.53464527, 0.02593486059, 9.81384723, -27.49023508];
c:=0.5:cl:=0.0:
IVP2:={x''(t)=-r1*sqrt(x'(t)^2+y'(t)^2)*(c*x'(t)+cl*y'(t)),
y''(t)=-g+r1*sqrt(x'(t)^2+y'(t)^2)*(cl*x'(t)-c*y'(t)),
x(0)=x0,x'(0)=vx0,
y(0)=y0,y'(0)=vy0}:
//Liste der Loesungen("fields"genannt)
fields:=[x(t),y(t),x'(t),y'(t)]:
ivp2:=numeric::ode2vectorfield(IVP2, fields):
Y2 := numeric::odesolve2(ivp2): Y2(6.00588);//t=6.00588s
//[75.9061706, -0.0001408615829, 7.23749789, -23.97649682];
//Animation
dt:=0.05:imax1:=200:imax2:=180:
plot(
plot::Point2d(Y1(t)[1], Y1(t)[2],
Color = RGB::Blue,
VisibleFromTo = t..t + 0.99*dt,
PointSize = 2*unit::mm
) $ t in [i*dt $ i = 0..imax1],
plot::Point2d(Y2(t)[1], Y2(t)[2],
Color = RGB::Green,
VisibleFromTo = t..t + 0.99*dt,
PointSize = 2*unit::mm
) $ t in [i*dt $ i = 0..imax2],
plot::Line2d([Y1(t - dt)[1], Y1(t - dt)[2]],
[Y1(t)[1], Y1(t)[2]], Color = RGB::Red,
VisibleAfter = t)
```

```
$ t in [i*dt $ i = 1..imax1],
plot::Line2d([Y2(t - dt)[1], Y2(t - dt)[2]],
[Y2(t)[1], Y2(t)[2]], Color = RGB::Blue,
VisibleAfter = t)$ t in [i*dt $ i = 1..imax2],
ViewingBox= [0..100,0..60])://x von 0 bis 100,y von 0 bis 60
```



Fig.2.5-3

Das Programm ist leider etwas "länglich", aber es lohnt sich sehr, sich ein wenig damit zu beschäftigen. Ich habe versucht, die Hauptpunkte durch kurze Kommentare zu erläutern. Sie sehen, dass die rote Trajektorie einen kleineren C-Wert hat, nämlich 0.3, als die blaue, die C = 0.5 hat. Bei beiden ist C_L = 0. Numerische Werte von $x, y, \dot{x}, \dot{y}$ erhalten wir für verschiedene Zeiten durch Einsetzen der Zeit in **Y1** und **Y2.**

Um Bahnen für zwei C-Werte gleichzeitig zu zeichnen, müssen wir das System (2.5-3) auch zweimal lösen, als IVP1 und IVP2. **IVP** = **I**nitial **V**alue **P**roblem (Anfangswert Problem), man kann jedoch einen anderen Namen wählen. Der Vektor (Liste) **Y** in `Y := numeric::odesolve2(ivp)`wird die Rechenergebnisse enthalten, d.h. Y(t)[1] = x(t), Y(t)[2] = y(t), Y(t)[3] = x'(t) und Y(t)[4] = y'(t).

Im Graphen verbinden wir im Zeitpunkt t den Punkt (x(t-dt),y(t-dt)) mit dem folgenden Punkt (x(t),y(t)) durch eine kurze Linie. Das Zeitinkrement dt und die maximale Anzahl der Punkte steht in der 1. Zeile der Animation. Wir benötigen i. Allg. verschieden viele Punkte für die einzelnen Kurven.

Wir werden das vorliegende Programm noch öfter benutzen, z.B. bei den Planetenbahnen. Es ist daher gut, sich durch Überlegen und Probieren mit dem Programm anzufreunden, z.B. auch dadurch, dass Sie zunächst nur eine Kurve darstellen, denn dadurch wird die Programmzeilenzahl stark reduziert. In `VisibleFromTo = t..t + 0.99*dt` wird dafür gesorgt, dass jeder Punkt nur kurze Zeit sichtbar ist. Der Faktor 0.99 sorgt dafür, dass man nicht zwei Punkte gleichzeitig sieht.

Wir wollen zum Schluss auch noch den **Drall** berücksichtigen, der zu sehr interessanten Bahnen führen kann. Wir wählen C = 0.3 und zwei Werte für C_L: 0.8 und 1.0. Ferner seien $v_0 = 60m/s, \alpha = 10°, R = 0.041m,\ m = 0.15\ kg$.Aus der Figur 2.5-4 erkennen wir, dass diese "Spinwerte" zwar sehr übertrieben sind, aber sie zeigen lustige Effekte.



Fig. 2.5-4

2.5.2 Tennisball und Flugzeug

Wir wollen annehmen, dass sich ein Tennisball von links nach rechts in der x-y-Ebene bewegt. Eine dünne Luftschicht haftet an der Oberfläche und wird sich mit dem Ball drehen, falls er einen Drall besitzt. Wenn er sich im Gegenuhrzeiger-Sinn dreht, wird die Luft sich auf der oberen Seite des Balls schneller bewegen als unterhalb. Dies hat zur Folge, dass der Luftdruck auf der Oberseite kleiner ist als unterhalb. Infolgedessen erfährt der Ball eine nach oben wirkende Kraft, die sogenannte Magnus-Kraft. (Regel: Je größer die Geschwindigkeit, desto kleiner der Druck. Die Kraftlinien sind auf der Oberseite des Balles dichter als auf der Unterseite.) *Gustav Magnus* erforschte 1835 die erwähnte Kraftwirkung und gab eine Erklärung, die auch heute noch akzeptiert wird.

Das Profil der Tragfläche eines Flugzeugs ist so gestaltet, dass die Luftgeschwindigkeit auf der Oberseite größer ist als auf der Unterseite. Es ensteht an der Unterseite der Tragfläche ein größerer Druck als an der Oberseite. Das Ergebnis ist eine resultierende Hubkraft. Wenn die Geschwindigkeit des Flugzeugs unter einen gewissen kritischen Wert fällt, verschwindet die Hubkraft, und das Flugzeug wird abstürzen, wenn es dem Piloten nicht gelingt, durch ein geeignetes Manöver die Hubkraft wieder zu aktivieren. Dazu hat er "nur" dafür zu sorgen, dass die Druckdifferenz zwischen Ober- und Unterseite sich wieder aufbaut.

Schon 1910 sagte der englische Physiker *J.J. Thomson* (The dynamics of a golfball. Nature 85, 1910, S. 2151-2157): *The spinning golfball is, in fact,* *a very efficient heavier-than-air flying machine; the lifting force may be many times the weight of the ball.*

Thomson benutze für seine Demonstrationen die Lorentz-Kraft auf Elektronen, die sich in einem Magnetfeld bewegen. (Die Lorentz-Kraft auf ein bewegtes Elektron ist $\vec{F} = -e\vec{v} \times \vec{B}$.)